Exercise 21: Autonomous Control
The code below is available for you to use in Project 2.  Study the code and try to understand how it works.  Install the sonar, QTI line sensors, IR detectors, and IR emitters (also install the speaker and LED if desired) on your boe-bot per the schematics in exercises 19 and 20.  Load the program below into the Basic Stamp Editor and modify it to match your component installations.  

Be sure to incrementally test each component---establish that all components are working properly.  This can be done by including the appropriate DEBUG statements and commenting out all subroutine calls except those needed to exercise the component being tested.

' {$STAMP BS2}

' {$PBASIC 2.5}

' -----[ I/O Definitions ]------------------------------------------------

Ping            PIN      0                      ' sonar pin
RMotor          PIN      14                     ' left servo motor

LMotor          PIN      15                     ' right servo motor

LLineSnsPwr     PIN      10                     ' left line sensor power
RLineSnsPwr     PIN       2                     ' right line sensor power
LLineSnsIn      PIN      12                     ' left line sensor input

RLineSnsIn      PIN       3                     ' right line sensor input

LfIrOut         PIN      13                     ' left IR LED output

LfIrIn          VAR    IN11                     ' left IR sensor input

RtIrOut         PIN       8                     ' right IR LED output

RtIrIn          VAR     IN6                     ' right IR sensor input

Speaker         PIN      4                      ' piezo speaker

StartLED        PIN      5                      ' display start delay

' -----[ Constants ]-------------------------------------------------------
Trigger       CON       5                  ' trigger pulse = 10 uS

Scale         CON    $200                  ' raw x 2.00 = uS

RawToIn       CON     889                  ' 1 / 73.746 (with **)

IsHigh        CON       1                  ' for PULSOUT

IsLow         CON       0

DistThresh      CON       36                    ' distance threshold

LFwdFast        CON     1000                    ' left motor forward; fast

LFwdSlow        CON      800                    ' left motor forward; slow

LStop           CON      750                    ' left motor stop

LRevSlow        CON      700                    ' left motor reverse; slow

LRevFast        CON      500                    ' left motor reverse; fast

RFwdFast        CON      500                    ' right motor forward; fast

RFwdSlow        CON      700                    ' right motor forward; slow

RStop           CON      750                    ' right motor stop

RRevSlow        CON      800                    ' right motor reverse; slow

RRevFast        CON     1000                    ' right motor reverse; fast

' -----[ Variables ]------------------------------------------------------
samples
   VAR     Byte                    ' # of samples for averaging

rawDist         VAR     Word                    ' raw distance measurement

inches          VAR     Word                    ' hold distance in inches

lightLeft       VAR     Word                    ' left light sensor reading

lightRight      VAR     Word                    ' right light sensor reading

rightThresh     VAR     Word                    ' right black threshold 
leftThresh      VAR     Word                    ' left black threshold 
irLeft          VAR     Word                    ' left IR sensor reading

irRight         VAR     Word                    ' right IR sensor reading

temp            VAR     Byte                    ' loop index

' -----[ Initialization ]--------------------------------------------------
' Sets black threshold to 1/4 the average of the two sensor readings.

' SumoBot must be placed over black playing surface before this code runs.

   Set_Threshold:                                  ' set photoresistor black                                                 
     leftThresh = 0

     rightThresh = 0
  FOR samples = 1 TO 5  ' take five samples

 GOSUB Read_Line_Sensors

 leftThresh = leftThresh + (lightLeft/5)

 rightThresh = rightThresh + (lightRight/5)
  NEXT
  leftThresh = leftThresh – (leftThresh/10)       ' set thresholds
  rightThresh = rightThresh – (rightThresh/10)
GOSUB Read_Line_Sensors
'  DEBUG CR, "rightThresh "

'  DEBUG DEC rightThresh, CR

'  DEBUG CR, "lefThresh "

'  DEBUG DEC leftThresh, CR

Start_Delay:                                    ' mandatory five second delay

  FOR temp = 1 TO 5

    HIGH StartLED                               ' show active

    PAUSE 900

    LOW StartLED                                ' blink each second

    FREQOUT Speaker, 100, 2500, 3000            ' beep each second

  NEXT

' -----[ Main Code ]-------------------------------------------------------
Do

  GOSUB Read_Line_Sensors

  IF (lightLeft < leftThresh) AND (lightRight < rightThresh) THEN

    GOSUB About_Face                            ' boundary dead ahead

  ELSEIF (lightLeft < leftThresh) THEN

    GOSUB Spin_Right                            ' boundary to the left

  ELSEIF (lightRight < rightThresh) THEN

    GOSUB Spin_Left                             ' boundary to the right

  ELSE

    PULSOUT LMotor, LFwdFast

    PULSOUT RMotor, RFwdFast

    GOSUB Search_For_Opponent

  ENDIF

Loop

' -----[ Subroutines ]-----------------------------------------------------
Search_For_Opponent:

  GOSUB Read_IR_Sensors

  GOSUB Read_Sonar

  IF (inches < DistThresh) THEN

    GOSUB Lunge                                 ' bot dead ahead

  ELSEIF (irLeft = 0) THEN

    GOSUB Spin_Left                             ' bot to the left

  ELSEIF (irRight = 0) THEN

    GOSUB Spin_Right                            ' bot to the right

  ELSE

    PULSOUT LMotor, LFwdFast

    PULSOUT RMotor, RFwdFast

  ENDIF

RETURN

' --[ Read Sensors ]--

Read_Line_Sensors:

  HIGH LLineSnsPwr  
  HIGH LLineSnsIn   
  PAUSE 1
  RCTIME LLineSnsIn, 1, lightLeft               ' read left sensor

  LOW LLineSnsPwr  
  HIGH RLineSnsPwr                            

  HIGH RLineSnsIn
  PAUSE 1
  RCTIME RLineSnsIn, 1, lightRight              ' read right sensor
  LOW RLineSnsPwr  

'  DEBUG CLS

'  DEBUG "right = ", DEC lightRight, CR

'  DEBUG "left = ", DEC lightLeft, CR

RETURN

Read_Sonar:

     rawDist = 0  

     Ping = IsLow                              ' make trigger 0-1-0

     PULSOUT Ping, Trigger                     ' activate sensor

     PULSIN Ping, IsHigh, rawDist              ' measure echo pulse

     rawDist = rawDist */ Scale                ' convert to uS

     rawDist = rawDist / 2                     ' remove return trip
     inches = rawDist ** RawToIn               ' convert to inches

RETURN

Read_IR_Sensors:

  FREQOUT LfIrOut, 1, 38500                     ' modulate left IR LED

  irLeft = LfIrIn                               ' read input (1 = target)

  FREQOUT RtIrOut, 1, 38500                     ' modulate right IR LED

  irRight = RtIrIn

'  DEBUG "left IR = ", BIN irleft, CR

'  DEBUG "right IR = ", BIN irright, CR

RETURN

' --[Opponent Chase]--

Lunge:                                          ' locked on -- go get him!

  FOR temp = 1 TO 15

    PULSOUT LMotor, LFwdFast

    PULSOUT RMotor, RFwdFast

    PAUSE 20

  NEXT

  PULSOUT LMotor, LFwdFast

  PULSOUT RMotor, RFwdFast

RETURN

Follow_Right:                                   ' spin right, fast

  PULSOUT LMotor, LFwdFast

  PULSOUT RMotor, RRevSlow

RETURN

Follow_Left:                                    ' spin left, fast

  PULSOUT LMotor, LRevSlow

  PULSOUT RMotor, RFwdFast

RETURN

' --[ Border Avoidance ]--

Spin_Left:                                      ' right sensor was active

  FOR temp = 1 TO 20

    PULSOUT LMotor, LRevFast

    PULSOUT RMotor, RFwdFast

    PAUSE 20

  NEXT

  PULSOUT LMotor, LFwdFast

  PULSOUT RMotor, RFwdFast

RETURN

Spin_Right:                                     ' left sensor was active

  FOR temp = 1 TO 20

    PULSOUT LMotor, LFwdFast

    PULSOUT RMotor, RRevFast

    PAUSE 20

  NEXT

  PULSOUT LMotor, LFwdFast

  PULSOUT RMotor, RFwdFast

RETURN

About_Face:                                     ' both sensors on Shikiri line

  FOR temp = 1 TO 10                            ' back up from edge

    PULSOUT LMotor, LRevFast

    PULSOUT RMotor, RRevFast

    PAUSE 20

  NEXT

  FOR temp = 1 TO 30                            ' turn around

    PULSOUT LMotor, LFwdFast

    PULSOUT RMotor, RRevFast

    PAUSE 20

  NEXT

  PULSOUT LMotor, LFwdFast

  PULSOUT RMotor, RFwdFast

RETURN

1.  Describe the modifications that you made to the program above in order to get it to run on your bot.










